
Example of a Type 1 Exit Program (RPGLE)

     ** E D V  -  B e r a t u n g  G m b H                                    **
    **                                                                       **
     **                     Systemberatung                                    **
     **                  IBM System AS/400                                    **
     **           Individualprogrammierung                                    **
     **                                   menten GmbH                         **
     **                                   Hauptstrasse 136 - 140              **
     **                                   51465 Bergisch Gladbach             **
     **                                   Telefon  +49 2202 2399 0            **
     **                                   Telefax  +49 2202-2399 23           **
     **                                   Email    info@menten-gmbh.de        **
     **                                   Homepage http://www.menten-gmbh.de  **
     **-----------------------------------------------------------------------**
     **                                                                       **
     **  PROGRAMM     : EXITREN                                               **
     **                                                                       **
     **  BESCHREIBUNG : Sample ‘TYPE 1‘ CVTSPL exit program for rename        **
     **                 streamfile -> CustomerNr.pdf                          **
     **                                                                       **
     **  DATUM        : Aug. 2002                                             **
     **                                                                       **
     **  AUTOR        : Ilja Kolsow                                           **
     **-----------------------------------------------------------------------**    
     **  Änderungsdatum :                                                     **
     **  Änderungsgrund :                                                     **
     **  Autor          :                                                     **
     **-----------------------------------------------------------------------**
     ** PROTOTYPE FOR STANDARD EXIT PROGRAM PARAMETERS                        **
     **-----------------------------------------------------------------------**
     D Parameters      PR                  ExtPgm(‘EXITREN‘)
      * Spooled file name
     D   iaSplFile                   10A
      * Name of the job which created the spooled file
     D   iaSplJob                    10A
      * User id of the job which created the spooled file
     D   iaSplUser                   10A
     * Job number of the job which created the spooled file
     D   iaSplJobNo                   6A
      * Spooled file number of the spooled file
     D   iiSplNbr                    10I 0
      * Spooled file user data
     D   iaUsrDta                    10A
      * Name of the stream file created
     D   iaStmFile                  128A
      * IFS directory in which stream file was created
     D   iaDir                      256A
      * Exit program parameter string extracted from spooled file
     D   iaExitParm                1024A
     **
     **-----------------------------------------------------------------------**
     ** PROTOTYPE FOR QCMDEXC                                                 **
     **-----------------------------------------------------------------------**
     **
     D Command         PR                  ExtPgm(‘QCMDEXC‘)
     D   iaCommand                32767A   CONST OPTIONS(*VARSIZE)



     D   inCmdLen                    15P 5 CONST
     **-----------------------------------------------------------------------**
     ** STANDARD EXIT PROGRAM PARAMETERS                                      **
     **-----------------------------------------------------------------------**
     D Parameters      PI
      * Spooled file name
     D   iaSplFile                   10A
      * Name of the job which created the spooled file
     D   iaSplJob                    10A
      * User id of the job which created the spooled file
     D   iaSplUser                   10A
      * Job number of the job which created the spooled file
     D   iaSplJobNo                   6A
      * Spooled file number of the spooled file
     D   iiSplNbr                    10I 0
      * Spooled file user data 
     D   iaUsrDta                    10A
      * Name of the stream file created
     D   iaStmFile                  128A
      * IFS directory in which stream file was created
     D   iaDir                      256A
      * User-definable exit program parameter.
     D   iaExitParm                1024A  
     **
     D laOldName       S           1024A
     D laNewName       S           1024A
     **
     D QUOTE           C                   ‘‘‘‘
     **
     D i               S             10I 0
     **-----------------------------------------------------------------------**
     ** main procedure                                                        **
     **-----------------------------------------------------------------------**
  **
      * If customer number found and passed as parameter, rename stream file
     C                   IF        iaExitParm <> *BLANKS
     **
      * Set up full path name to stream file
     C                   IF        iaDir = *BLANKS
     C                   EVAL      laOldName = QUOTE
     C                                       + %trim(iaStmFile)
     C                                       + QUOTE
     C                   ELSE  
     C                   EVAL      laOldName = QUOTE
     C                                       + %trim(iaDir)
     C                                       + ‘/‘
     C                                       + %trim(iaStmFile)
     C                                       + QUOTE
     C                   ENDIF
     **
      * Set up new path name for file using customer number in parameter
     C                   EVAL      laNewName = QUOTE
     C                                       + %trim(%subst(iaExitParm:1:10))
     C                                       + ‘.pdf‘
     C                                       + QUOTE
     **
      * Rename the stream file
     C                   CALLP(E)  Command(  ‘REN ‘
     C                                     + laOldName
     C                                     + ‘ ‘
     C                                     + laNewName:2053)
     **
     C                   dow       %error
     **
     C                   EVAL      i = i + 1
      * Set up new path name for file using customer number in parameter
     C                   EVAL      laNewName = QUOTE
     C                                       + %trim(%subst(iaExitParm:1:10))
     C                                       + ‘_‘ + %trim(%editc(i:‘Z‘))
     C                                       + ‘.pdf‘



     C                                       + QUOTE
      * Rename the stream file
     C                   CALLP(E)  Command(  ‘REN ‘
     C                                     + laOldName
     C                                     + ‘ ‘
     C                                     + laNewName:2053)
     C                   EndDo
     C                   ENDIF
     **
     C                   RETURN


